

Computing Line Integrals

© 2006 Mike May, S.J.- maymk@slu.edu

>
When working with line integrals, we will use the Student[VectorCalculus] package. It has nice commands to evaluate line integrals. To help our understanding, it seems worthwhile to go through using Maple to set up and evaluate line integrals over a parameterized curve in a step by step process, labeling all the intermediate steps. We will then produce new procedures that do everything in one command. We then note the Student[VectorCalculus] package commands that give the answer directly.

We also want to set a technical condition, specifying that the vector fields will be written as vectors rather than as linear combinations of basis vectors.

```
> restart;  
with(Student[VectorCalculus]);  
BasisFormat(false);  
[&x, `*`, `+`, `-`, `.`; <, >, <|>, BasisFormat, Binormal, CrossProduct, Curl, Curvature, (1)  
D, Del, Divergence, DotProduct, FlowLine, Flux, GetCoordinates, Gradient, Laplacian,  
LineInt, MapToBasis, Nabla, Norm, Normalize, PathInt, PositionVector,  
PrincipalNormal, RadiusOfCurvature, RootedVector, ScalarPotential, SetCoordinates,  
SpaceCurve, SpaceCurveTutor, SurfaceInt, TNBFrame, TangentVector, Torsion, Vector,  
VectorField, VectorFieldTutor, VectorPotential, diff, evalVF, int, limit, series]
```

A step by step example

First we need to define a vector field that we want to integrate, the parameterized path we are integrating over, and the limits on the parameter. For our example we will use the familiar circular vector field and integrate over half an ellipse.

```
> F := <-y, x>;  
VF := VectorField(F);  
C := <3*cos(2*t), 2*sin(2*t)>;  
trange := t=0..Pi/2;
```

$$F := \begin{bmatrix} -y \\ x \end{bmatrix}$$

$$VF := \begin{bmatrix} -y \\ x \end{bmatrix}$$

$$C := \begin{bmatrix} 3 \cos(2t) \\ 2 \sin(2t) \end{bmatrix}$$

$$trange := t = 0 .. \frac{1}{2} \pi$$


The process of integrating the line integral over the parameterized curve can be broken into 6 steps:

- 1) Plot a graph of the vector field and the parameterized curve.
- 2) Substitute the parameterization of the path into the field to make the field a vector valued function of the parameter.
- 3) Take the derivative of the path with respect to the parameter.
- 4) Evaluate the dot product and simplify. This reduces the problem to an integral over an interval.
- 5) Set up the line integral with the integrand being the dot product of the vector field and the derivative of the parameterized path.
- 6) Evaluate the integral.

We are ready to walk through the steps one at a time. As we work through the example with Maple, you should make sure that you can also do each step by hand.

The first step is to look at a graph of the vector field and the parameterized curve.

```
> P1 := VectorField(F, output=plot, view=[-4..4, -1..4], color=red);  
P2 := SpaceCurve(C, trange, color=blue);  
plots[display]({P1, P2}, axes=boxed, scaling =constrained);
```



The second step is to replace x and y and the vector field with the parameterizations of x and y at the appropriate point on the path.

```
> FonC := evalVF(VF, C);
```

$$FonC := \begin{bmatrix} -2 \sin(2t) \\ 3 \cos(2t) \end{bmatrix} \quad (1.1)$$

The third step is to take the derivative of the parameterized path.

```
> R := diff(C,t);
```

$$R := \begin{bmatrix} -6 \sin(2t) \\ 4 \cos(2t) \end{bmatrix} \quad (1.2)$$

The fourth step is to evaluate the dot product in the integrand and simplify.

```
> integrand1 := FonC.R;
   integrand := simplify(integrand1);
   print("The integrand is ", integrand);
      integrand1 := 12 sin(2 t)^2 + 12 cos(2 t)^2
      "The integrand is ", 12
```

The next step is to set up the line integral

```
> Int(integrand, trange);
```

$$\int_0^{\frac{1}{2}\pi} 12 dt \quad (1.3)$$

We now evaluate the integral and evaluate the result numerically, if needed.

```
> intval := int(integrand, trange);
   evalf(intval);
      intval := 6 pi
      18.84955592
```

An automated approach

For convenience we block the code into two procedures, one for plotting, and one for setting up the integral and evaluating it. This allows us to modify the problem with a minimal amount of work.

```
> pathplot := proc(vecfield, path, trange, window)
   local vfieldplot, parampath;
   vfieldplot := VectorField(vecfield,output=plot,view=
window);
   parampath := SpaceCurve(path,trange, color=black):
   plots[display]({vfieldplot, parampath});
end:
lineintegral:= proc(vecfield, path, trange)
   local intval, paramfield, dpath, integrand;
   print("the vector field is ", vecfield);
   print("the path is ", path, " with ", trange);
   print(Int((vecfield.Diff(path,t)), trange));
   paramfield := evalVF(VectorField(vecfield), path);
   print("the function on the path is ", paramfield);
   print(Int((paramfield.Diff(path,t)), trange));
   dpath := diff(path, t);
   print("the derivative of the path is ", diff(path, t));
   print(Int((paramfield.dpath), trange));
   integrand := simplify(paramfield.dpath):
   print("The integrand is ", integrand);
   print(Int(integrand, trange));
```

```

intval := int(integrand, trange);
print("the integral is ", intval);
print(evalf(intval));
end:

```

Thus, we can get results by entering the vectorfield and path and executing the two commands.

```


> SetCoordinates( cartesian[x,y] ):
vfield := <-y,x>;
path := <1+3*cos(t), -4+2*sin(t)>;
trange := t=0..Pi;
pathplot(vfield, path, trange, [-5..5, -5..5]);
lineintegral(vfield, path, trange);

```

$$vfield := \begin{bmatrix} -y \\ x \end{bmatrix}$$

$$path := \begin{bmatrix} 1 + 3 \cos(t) \\ -4 + 2 \sin(t) \end{bmatrix}$$

$$trange := t = 0 \dots \pi$$



"the vector field is ", $\begin{bmatrix} -y \\ x \end{bmatrix}$

"the path is ", $\begin{bmatrix} 1 + 3 \cos(t) \\ -4 + 2 \sin(t) \end{bmatrix}$, " with ", $t = 0 \dots \pi$

$$\int_0^{\pi} \begin{bmatrix} -y \\ x \end{bmatrix} \cdot \left(\frac{\partial}{\partial t} \begin{bmatrix} 1 + 3 \cos(t) \\ -4 + 2 \sin(t) \end{bmatrix} \right) dt$$

"the function on the path is ", $\begin{bmatrix} 4 - 2 \sin(t) \\ 1 + 3 \cos(t) \end{bmatrix}$

$$\int_0^{\pi} \begin{bmatrix} 4 - 2 \sin(t) \\ 1 + 3 \cos(t) \end{bmatrix} \cdot \left(\frac{\partial}{\partial t} \begin{bmatrix} 1 + 3 \cos(t) \\ -4 + 2 \sin(t) \end{bmatrix} \right) dt$$

"the derivative of the path is ", $\begin{bmatrix} -3 \sin(t) \\ 2 \cos(t) \end{bmatrix}$

$$\int_0^{\pi} (-3(4 - 2 \sin(t)) \sin(t) + 2(1 + 3 \cos(t)) \cos(t)) dt$$


"The integrand is ", $-12 \sin(t) + 2 \cos(t) + 6$

$$\int_0^{\pi} (-12 \sin(t) + 2 \cos(t) + 6) dt$$

"the integral is ", $-24 + 6\pi$
 -5.15044408

It should be noted that if all we want is the final answer, the VectorCalculus package has a LineInt command that can be used to plot the curve and field, set up the integral, and give the final answer.

```
> LineInt(VectorField(vfield), Path(path, trange), output=plot);  
LineInt(VectorField(vfield), Path(path, trange), output=  
integral);  
LineInt(VectorField(vfield), Path(path, trange));
```



$$\int_0^{\pi} (-3(4 - 2 \sin(t)) \sin(t) + 2(1 + 3 \cos(t)) \cos(t)) dt$$

$$-24 + 6\pi$$

Exercises:

1) Evaluate the line integral where $F = [\ln(y), \ln(x)]$ and C is the curve given parametrically by $[2t, t^2]$ with $t=2..4$.

[>

2) Evaluate the line integral where $F = [2y, -\sin(y)]$ and C is the unit circle traced in a counterclockwise directions from the point $(1, 0)$.

[>

3) Evaluate the line integral where $F = e^x i + e^y j$ and C is the part of the ellipse $x^2 + 4y^2 = 4$ from the point $(0, 1)$ to the point $(2, 0)$ in a clockwise direction.

[>

Line Integrals in 3-D

If the vector field is in three dimensions, we need to slightly modify the lineintegral command. In that case we will not plot the curve.

```

> lineintegral3d:= proc(vecfield, path, trange)
  local intval, paramfield, dpath, integrand;
  print("the vector field ", vecfield);
  print("the path ", path, " with ", trange);
  print(Int((vecfield.Diff(path,t)), trange));
  paramfield := evalVF(VectorField(vecfield), path);
  print("the function on the path is ", paramfield);
  print(Int((paramfield.Diff(path,t)), trange));
  dpath := diff(path, t);
  print("the derivative of the path is ", diff(path, t));
  print(Int((paramfield.dpath), trange));
  integrand := simplify(paramfield.dpath);
  print("The integrand is ", integrand);
  print(Int(integrand, trange));
  intval := int(integrand, trange);
  print("the integral is ", intval);
  print(evalf(intval));
end:

```

Using this command we redo example 4 from the text. Note that we have to specify that we are changing the coordinate system. (To go back to 2D we will need use SetCoordinates(cartesian[x,y]):)

```

> SetCoordinates( cartesian[x,y,z] ):
vfield := <x, z, -x*y>;
path := <cos(t), sin(t), 2*t>;
trange := t=0..3*Pi;
lineintegral3d(vfield, path, trange);

```

$$vfield := \begin{bmatrix} x \\ z \\ -xy \end{bmatrix}$$

$$path := \begin{bmatrix} \cos(t) \\ \sin(t) \\ 2t \end{bmatrix}$$

$$trange := t = 0 \dots 3\pi$$

$$\text{"the vector field "}, \begin{bmatrix} x \\ z \\ -xy \end{bmatrix}$$

$$\text{"the path "}, \begin{bmatrix} \cos(t) \\ \sin(t) \\ 2t \end{bmatrix}, \text{" with "}, t = 0 \dots 3\pi$$

$$\int_0^{3\pi} \begin{bmatrix} x \\ z \\ -xy \end{bmatrix} \cdot \left(\frac{\partial}{\partial t} \begin{bmatrix} \cos(t) \\ \sin(t) \\ 2t \end{bmatrix} \right) dt$$

"the function on the path is ",
$$\begin{bmatrix} \cos(t) \\ 2t \\ -\cos(t)\sin(t) \end{bmatrix}$$

$$\int_0^{3\pi} \begin{bmatrix} \cos(t) \\ 2t \\ -\cos(t)\sin(t) \end{bmatrix} \cdot \left(\frac{\partial}{\partial t} \begin{bmatrix} \cos(t) \\ \sin(t) \\ 2t \end{bmatrix} \right) dt$$

"the derivative of the path is ",
$$\begin{bmatrix} -\sin(t) \\ \cos(t) \\ 2 \end{bmatrix}$$

$$\int_0^{3\pi} (-3\cos(t)\sin(t) + 2t\cos(t)) dt$$


"The integrand is ", $\cos(t) (-3\sin(t) + 2t)$

$$\int_0^{3\pi} \cos(t) (-3\sin(t) + 2t) dt$$

"the integral is ", -4
-4.

Once again, if all we want is the final answer, the VectorCalculus package has a LineInt command.

```
> LineInt(VectorField(vfield), Path(path, trange), output=plot)
;  
LineInt(VectorField(vfield), Path(path, trange), output=
integral);  
LineInt(VectorField(vfield), Path(path, trange));
```



$$\int_0^{3\pi} (-3 \cos(t) \sin(t) + 2 t \cos(t)) dt$$

-4

▼ **Exercise:**

4) Evaluate the line integral where $F = -y \mathbf{i} + x \mathbf{j} + 5 \mathbf{k}$ and C is the helix $[\cos(t), \sin(t), t]$ for $t=0..4\pi$.

